
Learning Goals
• Understand FPGA’s structure and functionalities
• Verilog basics

– Signal, vectors, constants
– Operators, modules
– Wire, register types

• Verilog coding styles: structural v.s. behavioral
• Continuous assignment
• Procedural assignment

– Blocking v.s. non-blocking assignment
• Understand Verilog testbenches

Course Plan

• Lab 1: Display data with 7-segment display
– Build circuit using gates

• Labs 2-4: Hardware design using Verilog
– … and deploy on FPGAs

• Lab 5: Assembly programming 
– … extension of Lab 4

Evolution of Design Abstractions

Results
(design productivity)

Effort
(CAD tool effort)
Less human effort

McKinsey S-Curve

Transistor-level entry

Gate-level entry

HDL (Verilog, VHDL)

?

[source: Keutzer]

Hardware Description Languages
• Hardware Description Language(HDL)

– A language for describing hardware designs

• Advantages of HDLs
– Efficiently code large, complex designs

• Program at a more abstract level than schematics
• More readable than schematics

– EDA/CAD tools automatically generate hardware 

• Industry standards
– Verilog
– VHDL
– SystemVerilog: a successor to Verilog, gaining 

popularity

• Reading: Chapter 4

HDL-based Design Flow: Overview
• HDL code written to describe hardware after 

brainstorming over a given problem

• HDL code, via testing scripts, can be simulated by 
simulation tools.

• HDL code is synthesized by sophisticated compiler 
tools into hardware circuits.

HDL-based Design Flow: Detailed Look

• Back-end differs by target technology
– FPGA, ASIC, full-custom

to specific
device e.g. FPGA

gradually more low-level 
representation of the circuit

for extensive 
functionality 
testing

The compilation tool is 
intelligent enough to 
parse high-level circuit 
descriptions, and can 
also simplify logic 
(recall K-Maps?)



FPGAs & Verilog Tutorial  8ENGRD 2300

FPGAs
• Field Programmable Gate Array [FPGA]

– is an Integrated Circuit (IC) that can be programmed in 
the field *after* manufacturing.

Prototyping board with 
peripherals and 
everything required to 
make the FPGA work

The platform to test/deploy
our designed hardware

FPGA Structure

[source: Brown/Vranesic DLD/Verilog Textbook]

Stores Truth Tables Flip-flop storage 
element 

Inside a Logic Block

I/O blocks connected to 
peripherals

Why FPGAs
• Parallel execution in hardware
• Reconfigurable unlimited no. of times
• Readily available boards so quick to develop
• HDL Code → Hardware with little manual effort

• Uses
– Accelerating/prototyping algorithms in hardware
– variety of fields like defense, medical, machine learning

FPGA Ecosystem
• Multiple vendors

… among others

We’ll use this in
our course

• Let’s look at a couple of Verilog programs mapped 
to FPGAs

Recall Lab 1
• 4-bit Input 
   to 7-Segment Decoder

• Digits 0-9

I3 I2 I1 I0 Digit
0 0 0 0 0
0 0 0 1 1
0 0 1 0 2
0 0 1 1 3
0 1 0 0 4
0 1 0 1 5
0 1 1 0 6
0 1 1 1 7
1 0 0 0 8
1 0 0 1 9

Each segment’s expression 
was derived using K-Maps

𝒂 = 𝑰! + 𝑰" % 𝑰# + 𝑰$ % 𝑰# + &𝑰$ % &𝑰#

7-Segment Decoder

• Extended to all 16 inputs

7 segments defined ON or 
OFF for all 4-bit inputs

(straightforward & intuitive)

Verilog code
For 7-Segment 
Decoder

Let’s compile this code and see the 
hardware mapping on the FPGA

4-input Look-up Table 
(Truth Table)
for a single segment
output

4-bit Binary Input
Truth tables 
for all 7 
segments

Purely combinational circuit

Gate Level Representation
of a single segment output

7 outputs connected 
to 7-segment display

4 inputs connected 
to switches



FPGAs & Verilog Tutorial  18ENGRD 2300

FPGA reads inputs 
from switches and  
determines the  
output at 7-seg LEDs

D Flip-Flop storage element
as no direct T Flip-Flop available

Truth Table Logic determines 
when to toggle/reset/retain 
the D Flip-Flop value based 
on inputs T and RESET

Notice the output feedback

Result of compiling a T Flip-Flop in 
Verilog (Prelab 2A)

Sequential circuit example

VERILOG HDL

Verilog
• Developed in the early 1980s by Gateway Design 

Automation (later bought by Cadence)

• Supports modeling, simulation, and synthesis

• We will use a subset of language features
– Synthesizable vs. non-synthesizable



FPGAs & Verilog Tutorial  22ENGRD 2300

Verilog HDL
• Verilog is not a computer programming language
• Verilog is not a computer programming language
• Verilog is not a computer programming language

• Verilog is a Hardware Description Language
– It describes hardware
– Things happen simultaneously / in parallel
– …. whereas software is sequential

Signal Values
• Verilog signals can have 1 of 4 values

– 0  Logical 0, or false
– 1  Logical 1, or true
– z  High impedance, floating
– x  Unknown logical value

• May be a 0, 1, z, in transition, or don’t cares.

You’ll be mostly concerned with Logical 0 and 1 only.

Actual hardware has levels 0 and 1 only.

X and Z values might arise in your simulation tool.

Vectors
• Multi-bit values are represented as bit vectors 

(grouping of 1-bit signals)
– Right-most bit is least significant
– Example

• input [7:0] byte1, byte2, byte3;

• Common operations on bit vectors
– Bit selection

• byte1[5:2] or Zbus[3:7]
– Concatenation

• {byte1,byte2}
– Bitwise Boolean operators

• byte1 & byte2

Constants

• Binary constants
– 8’b00000001
– 4’b0111

• Decimal constants
– 4’d10
– 8’d345
– 32’d65536

4’b1001

Decimal number representing bit width

Base (b, d, h, o)

Operators
• Bitwise Boolean operators 

~ NOT 
& AND 
^ Exclusive OR 
| OR

• Arithmetic operators
+ Addition 
– Subtraction 
* Multiplication 

/ Division 
% Modulus 

<< Shift left 
>> Shift right

Operate bit-by-bit on multi-bit signals

Verilog Program Structure
• System is a collection of 

modules
– Module corresponds to a 

single piece of hardware

• Declarations 
– Describe names and types of inputs and outputs
– Describe local signals, variables, constants, etc.

• Statements specify what the module does

declarations

statements

module

Verilog Program Structure
module V2to4dec( i0,i1,en,y0,y1,y2,y3 );
   input i0,i1,en;
   output y0,y1,y2,y3;
   wire noti0,noti1;

   not U1(noti0,i0);
   not U2(noti1,i1);
   and U3(y0,noti0,noti1,en);
   and U4(y1,     i0,noti1,en);
   and U5(y2,noti0,     i1,en);
   and U6(y3,     i0,     i1,en);
endmodule

Declarations

Statements

List of inputs and outputs

input [3:0] B;In Declarations, can have something like:
This would be a 4-bit bus, with individual wire names

B[0] (LSB), B[1], B[2], B[3] (MSB)

Structural Coding Style

Verilog 
Hierarchy

declarations

statements

declarations

statements

declarations

statements

declarations

statements

module A

Module B Module C

Module D

A module can 
instantiate
other modules inside, 
forming a module 
hierarchy

Akin to programming language ‘classes’ 
instantiating objects of other classes

Verilog Programming Styles
• Structural

– Textual equivalent of drawing a schematic
– Uses instance statements to instantiate other modules 

and uses wires to connect them

• Behavioral
– Specify what a module does with a high-level description
– Uses procedural code (e.g., in always blocks) or 

continuous assignment (i.e. assign) constructs 

We can mix the structural and behavioral styles
 in a Verilog design

More efficient - we’ll mostly use this style in our labs

You just saw this a couple of slides ago

Net and Variable Types
• Verilog provides multiple net and variable types

• We will main use two types

– wire: represents a physical connection (net) between hardware elements
• Used for structural style and continuous assignments
• Default if you do not specify a type
• Can only be used to model combinational logic
• Cannot be used in the left-hand side in an always block

– reg: a variable that can be used to store state (registers)
• Used in the procedural code (always blocks)
• May also be used to express combinational logic
• Can be used to model both combinational & sequential logic
• Cannot be used in the left-hand side of a continuous assignment statement

You must be clear on their usage!
They’re a common source of errors.

More on this later.

Internal Wires
• Sometimes it is useful to name internal signals and 

assign intermediate values
– Declaration:

wire temp_value;

– Assignment:

assign temp_value = some_statement;

Example: 2-to-1 MUX
• select = Selector signal
• If select = 0, out = x
• If select = 1, out = y

select x y out
0 0 0 0
0 0 1 0
0 1 0 1
0 1 1 1
1 0 0 0
1 0 1 1
1 1 0 0
1 1 1 1

out = select’ x + select y

Structural Style

The order of the gates instantiation does not matter.
Essentially describing the schematic textually.

tx

ty

module MUX2_1 (x, y, select, out);
  input  x, y;
  input  select;
  output out;

  wire inv_sel, tx, ty;
  
  NOT not0(inv_sel,select);
  AND and0 (tx, x, inv_sel);
  AND and1 (ty, y, select);
  OR  or0 (out, tx, ty);
  
endmodule

x

y

select

out

inv_sel

Internal wires 
used to connect 

gates.



FPGAs & Verilog Tutorial  35ENGRD 2300

Structural Practice Example 

F    

A    

B    
C    

module mycircuit(                      );
   input A, B, C;
   output F;

   nand U1(                 );

   ….

endmodule

Structural Practice Example

F    

A    

B    
C    

module mycircuit(A, B, C, F );
   input A, B, C;
   output F;
   wire temp;

   nand U1(temp, B, C);

   nand U2(F, A, temp);

endmodule

temp

Behavioral Coding Style

• An assign statement represents continuously 
executing combinational logic 

• Multiple continuous assignments happen in parallel; the 
order does not matter

module MUX2_1 (x, y, select, out);
  input  x, y;
  input  select;
  output out;
  
  assign out = (~select & x) | (select & y);
  
endmodule

Behavioral Coding Style: 
Continuous Assignments

Assign used to set 
values for wires

Output out is a wire 
by default.

Continuous Assignments
• An assign statement means continuously executing 

combinational logic
– LHS must be a wire type
– RHS is recomputed when a value in the RHS changes
– The new value of the RHS is assigned to the LHS

• Example:  
– assign y0 = en & ~i0 & ~i1;

Behavioral Coding Style:
Combinational Logic with Always Blocks

module MUX2_1 (x, y, select, out);
  input  x, y;
  input  select;
  output reg out;
  
  always @(x, y, select) 
  begin
    out = (~select & x) | (select & y);
  end
  
endmodule

• An always block is 
reevaluated whenever a 
signal in its sensitivity
list changes

• Formed by procedural 
assignment statements
– LHS inside the always 

block must be reg

Here, the signal is a ‘reg’ even though modeling combinational logic!
reg can be used to model both combinational and sequential logic.

This is Behavioral Style Too!
module MUX2_1 (x, y, select, out);
  input  x, y;
  input  select;
  output reg out;
  
  always @(x, y, select) 
  begin
 if (select == 0)
  out = x;
 else
  out = y;
 end
  
endmodule

• Closely resembles 
verbal circuit 
description!

• High-level and intuitive.

• Synthesis Tool is 
intelligent enough to be 
able to infer a mux!

Conditional Statement

Conditionals
• Logical operators used in conditional expressions

– &&  logical AND
– ||     OR
– !      logical NOT
– ==   logical equality
– !=    logical inequality
– >     greater than
– >=   greater than or equal
– <     less than
– <=   less than or equal

• Don’t confuse with Bitwise Boolean operators

Concurrent Elements/Blocks
module temp (…..);
  input  …; 
  output …;
  
 assign … = ...;

 always @(…) 
   begin  …
   end

 always @(…) 
   begin  …
   end
  
endmodule

• Always blocks execute 
concurrently with other 
always blocks, instance 
statements, and 
continuous assignment 
statements in a module

Sensitivity Lists in Always Blocks 
• Include every variable in the always block 

sensitivity list
– always @ (signal list)

• Simple alternative
– always @ (*)

• For edge-triggered behavior
– always @ (posedge/negedge SIGNAL_NAME)

Used to describe Combinational Logic 

Used to describe Sequential Logic 
More later!

Assignments in Verilog
• Continuous assignments apply to combinational 

logic only

• Always blocks contain a set of procedural 
assignments (blocking or non-blocking)
– Can be used to model either combinational or 

sequential logic
– Always blocks execute concurrently with other always 

blocks, instance statements, and continuous 
assignment statements in a module

Procedural Statements
• Procedural statements are similar to conventional 

programming language statements
– Begin-end blocks

•begin procedural-statement … procedural-statement end

– If
•if ( condition ) procedural-statement else procedural-statement

– Case
•case ( sel-expr ) choice : procedural-statement … endcase

– Blocking assignment
•variable name = expression ;

– Nonblocking assignment
•variable name <= expression ;

More later!

You saw this in the Behavioral example.

• So far you’ve seen how to implement 
combinational logic.
– output as a function of input
– determined using logic gates

• Now for sequential logic…
– introduce storage elements like latches, flip-flops

Sequential Logic in Always Blocks
reg Q;

always @( clk, D ) 
begin 

if ( clk ) 
Q <= D; 

end 

always @( posedge clk ) 
begin 

Q <= D; 
end 

• Sequential 
logic can only 
be modeled 
using always 
blocks

• Thus, Q must 
be declared as 
a “reg”

QD

C

QD

CLK

D latch

DFF

Flip-Flop with a reset?
reg q;

always @ (posedge clk or posedge reset)
    if (reset==1) 
        q <= 1’b0;
    else
        q <= d;

Rising clock edge
Asynchronous 

Reset
in between

clock edges

Flip-Flop assignment

Flip-Flop with a reset?
reg q;

always @ (posedge clk)
    if (reset==1) 
        q <= 1’b0;
    else
        q <= d;

Rising clock edge
Synchronous Reset

Flip-Flop assignment

Blocking Assignments
• Blocking assignments ( = )

– Simulation behavior: Right-hand side (RHS) evaluated 
sequentially; assignment to LHS is immediate

reg Y, Z;
always @ (posedge clk)
begin

Y = A & B;
Z = ~Y;

end
Blocking assignments

Simulator interpretation Resulting circuit (post synthesis)
Ynext = A & B
Znext = ~(A & B)

ZY

B
A

Y and Z are flip-flops 
in actual hardware



FPGAs & Verilog Tutorial  59ENGRD 2300

Nonblocking Assignments
• Nonblocking assignment ( <= )

– Simulation behavior: RHS evaluated in parallel (order doesn’t 
matter); Assignment to LHS is delayed until end of always block

reg Y, Z;
always @ (posedge clk)
begin

Y <= A & B;
Z <= ~Y;

end

Znext = ~Y  // reading the old Y
Ynext = A & B

Nonblocking assignments

ZY

B
A

Simulator interpretation Resulting circuit (post synthesis)

Y and Z are flip-flops 
in actual hardware

Rules to Follow
• Use blocking assignments to model combinational logic 

within an always block.
– or just implement combinational without an always block, using assign statements

• Use non-blocking assignments to implement sequential 
logic.

• Do not mix blocking and non-blocking assignments in the 
same always block.

• Do not make assignments to the same variable from more 
than one always block

VERILOG TESTBENCHES

To extensively test our HDL design using 
simulations before deploying it in hardware

We should have a comprehensive test case 
coverage (including edge cases)

Why Testbenches?
• Generate stimulus
• Monitor UUT (Unit Under Test) output

• Automation

Verilog Testbench

Verilog
UUT

UUT
Stimulus

UUT
Monitor

i.e. test inputs

and hope it matches the expected output

Testbenches vs. Hardware
• Important distinction

– Verilog testbenches DO NOT describe hardware (i.e., 
non-synthesizable)

– Verilog testbenches look like a software program

Verilog

Synthesizable
subset

e.g. we frequently use ‘delays’ 
in testing when applying new 
test inputs, and print out 
signal values in the output log

Clearly such features are only 
relevant to simulation, they 
CAN’T translate to hardware

Building Blocks of a Testbench
• Timescale
• Module definition
• UUT instantiation
• Stimulus generation
• Monitoring output

Timescale
• Specifies the simulation granularity
• Syntax

• Example

// sets the granularity at which we simulate
`timescale 1 ps / 1 ps

`timescale time_unit base / precision base
for fractional delayunit of delays

Test Module Definition
• A testbench is a module too!
• No inputs or outputs

module treg4bit_test();

Module Declarations
• Define input and output signals that will link to our 

UUT

// inputs
  reg         CLK50;
  reg         RESET;
  reg  [3:0]  IN;

  // outputs
  wire [3:0]  OUT;

UUT Instantiation
• Same as instantiating a submodule

// instance of treg4bit
  treg4bit UUT (
    .CLK(CLK50),
    .RESET(RESET),
    .IN(IN),
    .OUT(OUT)
  );

Verilog Testbench

Verilog
UUT

UUT
Stimulus

UUT
Monitor

Stimulus Generation: Clock
• Generate a 50MHz clock

always
 begin
    CLK50 = 1'b0;
    CLK50 = #10000 1'b1;
    #10000;
 end

#10000 means a delay i.e. wait 
for 10000 time units before 
changing the value of CLK50

The total time period is 20000 
time units

Stimulus Generation: Initial Block
• Use initial block to write sequential test cases for 

the circuit

initial
  begin

    // reset circuit
    ...

    // test cases
    ...

  end

Stimulus Generation: Reset
• At beginning of initial block, give initial values to the 

circuit

// reset circuit
RESET = 1'b1;
IN = 4'b0;

#20000;
// test whether reset is correct
...

// enable and start running
RESET = 1'b0;

Stimulus Generation: Set Inputs
• Set inputs to values for the test case

// enable toggling for only some of the flip-flops
IN = 4'b0011;

#20000;  // wait a cycle to observe result

Monitoring Output
• Compare actual output with expected output
• Display information

• Can also use waveforms to see inputs and outputs 
change over time

if(OUT == 4'b0) begin
      $display("MSIM> Reset is correct, OUT (value in 

register) now %4b", OUT);
    end
    else begin
      $display("MSIM> ERROR: Reset is incorrect. OUT 

(value in register) should be 0000, but is %4b", 
OUT);

    end

Testbench output i.e. 
print/display statements

Waveform display
Input/Output 
signals

Pause/Terminate simulation
• Pause simulation (can resume later)

• Terminate simulation

$stop;

$finish;

Learning Goals
• Understand FPGA’s structure and functionalities
• Verilog basics

– Signal, vectors, constants
– Operators, modules
– Wire, register types

• Verilog coding styles: structural v.s. behavioral
• Continuous assignment
• Procedural assignment

– Blocking v.s. non-blocking assignment
• Understand Verilog testbenches


